# Maintainer: Evangelos Foutras <evangelos@foutrelis.com>
# Contributor: AndyRTR <andyrtr@archlinux.org>
# Contributor: Abhishek Dasgupta <abhidg@gmail.com>

pkgname=xfce4-time-out-plugin
pkgver=1.0.2
pkgrel=2
pkgdesc="Take a break from your computer with this plugin for Xfce4"
arch=('x86_64')
url="http://goodies.xfce.org/projects/panel-plugins/xfce4-time-out-plugin"
groups=('xfce4-goodies')
license=('GPL')
depends=('xfce4-panel')
makedepends=('intltool')
source=(https://archive.xfce.org/src/panel-plugins/$pkgname/${pkgver%.*}/$pkgname-$pkgver.tar.bz2)
sha256sums=('c69fb7218c685e516454810291c5245c49a80d774a6bded59c645a1af427c4e5')

build() {
  cd "$srcdir/$pkgname-$pkgver"

  ./configure \
    --prefix=/usr \
    --sysconfdir=/etc \
    --libexecdir=/usr/lib \
    --localstatedir=/var \
    --disable-static \
    --disable-debug
  make
}

package() {
  cd "$srcdir/$pkgname-$pkgver"
  make DESTDIR="$pkgdir" install
}

# vim:set ts=2 sw=2 et:
